

Spice subcircuit accurately models zener characteristics

Steven M Sandler

My work often involves examining mission-critical circuits for reliability. Interestingly, for almost every circuit that comes through containing a zener diode an error exists in the circuit's analysis. It appears that although zeners seem simple, people really don't know how to characterize and properly use them. Indeed, zener characteristics change dramatically depending on operating conditions. Consider the most common errors in their application: first, operation at currents other than the specified zener current results in a shift in the nominal voltage and also is likely to affect device-temperature coefficient; second, improper understanding of the devices' dynamic impedance characteristics.

Furthermore, most Spice models are inadequate to handle these vagaries. Thus, at my company we found it necessary to develop a more detailed zener model, which this column describes.

Steve Sandler is the Vice President of Analytical Engineering Inc (Mesa, AZ (602) 890-7197, www.aeng.com, ssandler@aeng.com), a firm specializing in the design, analysis and simulation of electronic equipment for military and space apps. He is author of SMPS Simulation with Spice 3 published by McGraw Hill.

Poor device understanding

To understand why designers mistreat the zener diode, let's start by looking at some device basics. A zener conducts current in both directions: in Forward Conduction mode it appears as a standard silicon diode with a typical breakdown of approximately 0.7V; in Reverse Conduction mode, it conducts at the zener voltage.

In general practice, low-voltage (<12V) zeners are typically alloy-diffused junction devices, while higher voltage types are typically diffused-junction devices. Manufacturers use different processes for developing these junctions, and these differences can drastically alter device performance.

In addition, a zener's temperature coefficient is a function of the zener voltage and current. Devices rated below 5.6V typically have a negative tempco, and those rated above 5.6V typically have a positive tempco—but those rated very close to 5.6V can go either way. Manufacturers fabricate temperature-compensated devices by combining junctions with both positive and negative temperature coefficients. Thus, a typical 5.6V zener has a positive tempco of 2mV/°C, which manufacturers can offset by a typical silicon junction with a tempco of -2mV/°C—resulting in a 6.2V device with nearly zero tempera-





Fig 1—A simple test circuit helps assess how well Spice works with functions when used with its standard zener diode model.

ture coefficient. This process explains why most reference diodes are rated at approximately 6.2V.

Zeners follow a similar pattern with impedance. Devices just below 5.6V generally provide higher impedance than devices rated just above 5.6V. At voltages very close to 5.6V, they offer very low impedance in the range of tens of ohms—but the exact value again is unpredictable. This consideration is important in applications where ripple rejection is critical or where diode impedance can affect circuit stability.

In reviewing manufacturers data-sheets, you'll often find that they limit specs to a table that provides the



Lab data vs Spice primitive zener diode model			
Zener current (mA)	Zener voltage-measured (V)	Zener voltage-modeled (V)	Error (V)
0.025	3.58	4.24	0.66
0.250	4.29	4.30	0.01
5.0	4.73	4.38	0.35

Table 1--Depending on the zener current, the match between measured and modeled zener voltage varies quite a bit.

nominal zener voltage and tolerance, temperature coefficient and maximum zener impedance at the zener test current. In some cases, differences in manufacturing techniques can result in a zener impedance that's an order of magnitude or more lower than the maximum specified value. While this effect is beneficial in many apps, it can also be detrimental in others.

Poor simulator support

Not only do Spice simulators not handle these special cases, they generally have no primitive-level support for the zener diode at all. At the primitive level, engineers realize zener characteristics by using BV (breakdown voltage) and IBV (inverse breakdown voltage) characteristics of the Spice diode, corresponding to the avalanche characteristics or breakdown voltage and breakdown current. The relationship for this effect, in terms of reverse current (the zener current), is $I_R = IBV \cdot e^{((-V_D - BV) / VT)}$ where VT is the thermal voltage $K \cdot T / Q$, or 26 mV at room temperature.

Rearranging this relationship to solve for VD leads to

$$V_D(I_Z) = \left(\ln \left(\frac{I_Z}{IBV} \right) + \frac{1}{VT} \cdot BV \right) VT$$


and differentiating with respect to IZ results in

$$\frac{dV_D(I_Z)}{dI_Z} = \frac{1}{I_Z} \cdot VT$$

which is the device's dynamic impedance—and note that this method provides little control over device characteristics.

To demonstrate how well a Spice zener model built around standard diodes performs (or doesn't perform), the author arbitrarily selected a 1N4623, a 4.3V 250 μA zener, for a

Fig 2--An improved Spice subcircuit for the zener diode uses separate diode/source pairs for the full and soft regions plus a diode alone for the forward region.



series of tests. The author measured its reverse-current, forward-current and capacitance characteristics at different currents. He then simulated the same test circuit in Spice with the model in Fig 1. Table 1 compares measured and simulated values. Note that the zener voltage is correct at the zener test current (250 μA)—but if the current is something other than that value, the model shows poor accuracy.

An improved subcircuit

Despite these discrepancies, Spice is flexible enough to allow you to construct a subcircuit that accurately models zener performance at any current. The new model relies on the fact that a zener's characteristic curve (a plot of zener voltage vs current) breaks easily into two regions. One of them, called the soft region, is the portion of the curve well below the zener test current IZT. The other portion is called the forward region.

To handle this situation, a second Spice model representing the 1N4623

(Fig 2) uses a separate subcircuit for each region. Each subcircuit is based on the Spice representation for the forward conduction of a diode combined with a voltage source. The complete model combines three diodes and two voltage sources. Breaking it down into the three parts:

- diode DHIGH and voltage source V1 model the zener's full operating region.
- DLO and V3 account for variations in the soft region.
- DF (with no associated voltage source) accounts for variations in the forward region, and it also models the device's voltage-dependent capacitance characteristics.

The Spice code in Listing 1 defines the netlist for a test circuit based on this model. Note that the constant current source I1 represents the zener current.


To complete this model and test its validity, for each of the three subcircuits you must determine values for the constants that define the diodes and voltage sources that make up the Spice primitives. One way to estimate their values is to set up a system of equations based on the equality

$$V_{zener} = \ln\left(\frac{I_{zener} + IS}{IS}\right) \cdot N \cdot 0.026 + RS \cdot I_{zener}$$

which appears in many Spice references, and to solve for the unknowns based on laboratory measurements of zener voltage at several zener currents. For this task, Mathcad's Minerr function solved the equations; that function works like the Find function except it returns an approximate solution when Find would report that no solution exists.

Let's start with the soft region. This time, when dealing with the defining equation ignore the effects of RS, which are insignificant. Note that the author added RS = 1.5K in the Spice definition as a tweaking factor, de-

Fig 3--To measure zener capacitance, the author set up a test circuit and measured it at three different voltages.



termined by trial and error, to soften the corner between the soft and full regions. To use the Minerr function, you first supply initial guesses for the unknown variables, and the author chose the following as rough guesses: $V3 = 3.5$, $IS = 1 \times 10^{-7}$ and $N = 5$. Because you're solving for three unknowns, you need at least three simultaneous equations; select three voltage-current pairs taken from lab measurements in the soft region and plug the values of V_{zener} and I_{zener} directly into the above equation for each of the three pairs. Note that these representations also insert the known value of 0.026 for VT.

$$3.58 - V3 = \ln\left(\frac{0.000025 + IS}{IS}\right) \cdot 0.026N$$

$$3.86 - V3 = \ln\left(\frac{0.00005 + IS}{IS}\right) \cdot 0.026N$$

$$3.99 - V3 = \ln\left(\frac{0.000075 + IS}{IS}\right) \cdot 0.026N$$

Next apply the Minerr function to solve for the three unknowns:

$$\text{minerr}(N, V3, IS) = \begin{bmatrix} 14.589 \\ 1.916 \\ 3.101 \cdot 10^{-7} \end{bmatrix}$$

The model is complete, but how well does it perform? To find out, supply the model with each of the three measured values of zener current, note the predicted value of zener voltage and see how close these predictions match previously measured lab results. In this case $VZ_{\text{model}}(0.000025) = 3.586$, while $VZ_{\text{measured}} = 3.58$; $VZ_{\text{model}}(0.00005) = 3.846$, while $VZ_{\text{measured}} = 3.86$; and $VZ_{\text{model}}(0.000075) = 3.99$, while $VZ_{\text{measured}} = 3.999$. These results show excellent agreement with measured data.

With this submodel for the soft region complete, now repeat this process to determine the model for the full zener region. Note, though, that this time you do want to solve for RS and must provide an initial guess. To minimize the number of unknowns, in this step define a fixed value for $IS = 1 \times 10^{-7}$. Thus, the set of initial guesses is $N = 5$, $RS = 10$ and $V1 = 3.5$. Applying the Minerr function to a set of five measured pairs (increased from three to help produce better results) gives $N = 4.705$, $V1 = 3.339$ and $RS = 13.34$. Again, when you plug these values into the model and check it against laboratory values, you'll





Fig 4--The correlation between the enhanced zener-diode Spice model and measured data is excellent for both the zener region (a) and the forward region (b).

find they're in good agreement.

Finally apply the procedure to characterize operation in the forward region. Here there's no voltage source, so the three unknowns for which you must supply an initial guess are $N = 1$, $IS = 3.8 \times 10^{-12}$ and $RS = 5$. The author set up a system of seven simultaneous equations, the redundancy helps Minerr achieve a better fit, and the software determined that $N = 1.18$, $IS = 1.76 \times 10^{-13}$ and $RS = 1.638 \times 10^{-3}$. Once again, the resulting model operates in good agreement with measured data.

A final characteristic of concern as described earlier is diode capacitance, which can become critical in hf cir-

cuits. Indeed, depending on voltage, even a zener's capacitance can vary by thousands of picoFarads.

Thus, the author first measured diode capacitance at three voltages in the lab with the setup in Fig 3. Using an HP3577A network analyzer he extracted the 3-dB gain at each of the three datapoints and then calculated capacitance with the equation

$$CJ = (1 / 2\pi \cdot 990 \cdot F3db) - C_{probes}$$

He obtained probe capacitance by measuring circuit capacitance with the zener removed. He then entered the resulting datapoints into Mathcad and derived the terms for CJO , VJ , and M using the Minerr function to solve a system of equations based on the Spice

representation of junction capacitance:

$$C_{junction} = CJO \left(1 + \frac{V_{test}}{V_j} \right)^{-M}$$

After setting initial guesses to $M = 0.3$, $CJO = 100 \times 10^{-12}$ and $V_j = 0.75$, the software generated the following model values: $CJO = 2.963 \times 10^{-10}$, $V_j = 0.547$ and $M = 0.294$ —which once again generate a model in good agreement with the measured data.

Now enter all of the derived parameters into the subcircuit of Fig 2, and you end up with a zener subcircuit for Spice as given in Listing 2. A comparison of the derived Spice subcircuit and measured data appear in Fig 4, and the correlation is excellent for both the zener region and the forward region. Listing 2 shows the new Spice model for the 1N4623, and it accurately represents the device's behavior—but note that it does so only at room temperature. PE&IN

Listing 1: Test-circuit netlist for Fig 2

```
.MODEL DHIGH D IS=1E-7 RS=9.11
N=4.989
.MODEL DLO D IS=3.1E-7 RS=1.5K
N=14.589
.MODEL DF D IS=1.76E-13 RS=1.638M
N=1.18 CJO=296.3P VJ=.547
M=.294
.OPTIONS TEMP=25
.OP
V1 1 0 3.281
D2 2 3 DLO
V3 3 0 1.916
I1 0 2 250U
D3 0 2 DF
D1 2 1 DHIGH
.END
```

Listing 2: 1N4623 Spice subcircuit

```
.SUBCKT ZN4623 4 2
.MODEL DHIGH D IS=1E-7 RS=9.11
N=4.989
.MODEL DLO D IS=3.1E-7 RS=1.5K
N=14.589
.MODEL DF D IS=1.76E-13 RS=1.638M
N=1.18 CJO=296.3P VJ=.547
M=.294
V1 1 4 3.281
D2 2 3 DLO
V3 3 4 1.916
D3 4 2 DF
D1 2 1 DHIGH
.ENDS
```

Editorial Feedback

This article's value to me was:
High—263 Average—264 Low—265